
News 
Our pre-Neuroscience training days this year will take place on Thursday 10th and Friday 11th of 
November at the Washington Plaza Hotel, Washington DC. Details of the training day sessions and a 
booking form can be found at www.ced.co.uk/training. 

The latest update for Spike2 is 5.13, which is available from the CED downloads page for registered v5 
users. This revision includes new commands to fit sinusoids and sigmoids to data. 

A full list of new features is available from the updates section of the website. 

A demonstration version of the current software is also available here. 

Version 4.23 of Spike2 is downloadable for registered v4 users. 

Version 3.04 of Signal is now available for registered v3 users. This revision includes extended telegraph 
support and logging of digital markers. 

A demonstration version of Signal version 3 is available here. 

Version 2.15 of Signal is downloadable for registered v2 users. 

Spike2  

Q. I would like to plot the mean blood pressure over the last 5 beats as a moving average. Is there a 
way to achieve this with Spike2? 

A. Using the Active Cursors, it is possible to take a wide range of measurements from channel data. 
These measurements can be plotted to a separate XY view or can be placed in a new channel 
within the data file. One benefit of plotting values directly to a data channel is the provision to take 
additional measurements based on this derived data to further channels in the same file.  

 Using this example we can generate a measurement channel to mark each BP peak, and then 
generate a second channel to give mean blood pressure over the last n beats. The result of this is 
shown below using the example Blood Pressure Waveform file available in the Spike5 data folder. 
Channel m1 shows the marked peaks, and channel m2 shows the Mean Blood Pressure value 
over the last 5 beats. 

 The first step is to generate the channel to mark the BP peaks. From the Analysis menu, select 
Measurements and then Data channel. In the new measure to channel dialog set the channel type 
to Event and set the Cursor 0 stepping mode to find the BP peaks. Remember that the setting for 
amplitude is the level that the waveform must fall by (in Y-axis units) after the peak to be marked. 

 

Settings to create a peak marker channel 

 Click New and then Process in the dialog. A new measurement channel is created to mark the 
pressure peaks as events.  



 To get the value of mean pressure over the last 5 beats, we need to set-up a second Active 
Cursor. Create a new cursor from the Cursor menu or click on the data file and press Ctrl+1. 
Select Active Modes… from the Cursor menu and set the Search method for Cursor(1) to Data 
Points, and the Search channel to be our event channel containing the marked peaks. Set the 
End position for search to be at least the equivalent time of 5 beats before Cursor (0), which will 
be a negative time. The Point count field is used to set the number of peak counts required. Using 
the settings detailed below, Cursor(1) will count back to find the 5th preceding peak marker within 
a 10 second time range. 

 

Settings for Cursor(1) to find 5th preceding beat 

 Finally from the Analysis menu, select Measurements and then Data channel. Because this final 
plot will be a measurement with a Y-value, set the Channel type to RealMark to enable the Y 
measurements field and set its Type to Mean. The Start time should be set to Cursor (1) and the 
End time to Cursor (0), as below.  



 

Signal 

Q. Is there provision in Signal to read telegraph data directly from the Axon MultiClamp 700 computer 
controlled patch-clamp amplifier? 

A. The latest version of Signal includes additional telegraph support for this amplifier. It allows Signal 
to read information such as changes to the amplifier gain settings, low-pass filter settings, 
membrane capacitance and series resistance from the MultiClamp command software.  

 During the installation, Signal gives you the option of installing telegraph support for the 
MultiClamp 700. Once installed, the Port Setup page of the sampling configuration will then 
enable the configuration dialog. This is then used to configure Signal so that it can communicate 
with the amplifier command software. 

    

 

 You select the type of amplifier and the ADC ports on the 1401 that are reading data. Below is a 
checkbox that tells Signal how to automatically scale and calibrate the data. The channel names, 



units and calibration values are read directly from the amplifier. This is the easiest way to use the 
telegraph data as it will automatically apply the settings from the amplifier setup. If you disable this 
option then Signal will use the specified channel name, units and calibration from the Port Setup 
page of the sampling configuration. 

      

Scripts: Spike2  

Q. I am recording ECG data and I would like to get Spike2 to emit a ‘beep’ for every R-wave event 
on-line, much like a standard ECG monitor would do. Is this possible? 

A. The attached script, ThresholdBeep.s2s emits a ‘beep’ in response to a threshold crossing on 
a nominated channel. It opens a new file ready to sample and prompts the user to specify the 
channel to monitor and set a threshold using a horizontal cursor. This threshold can be adjusted 
at any time during sampling. 

 

Scripts: Signal  

Q. It would be useful when recording to have an auditory signal for each sampled frame. It would be 
even better if you could have separate tones to signify different frame states during sampling. 

A. The attached script, SampleBeep.sgs plays a short tone for each sampled frame. The tone 
pitch changes depending on the frame state. 

Did you know…? 
You can set individual channel background colours in Spike2 and Signal data files, as well as specific view 
colours for data and XY views. You can also set independent background colours for result windows, the 
cluster background and WaveMark dialogs in Spike2. 

Recent questions  

Q. I am using the Spike2 video application to capture multimedia files alongside my data files. 
Because I record for up to 6 hours the multimedia files generated are rather large. To keep the 
size down, is there a way of pausing or stopping video capture during sampling, or a function that 
allows adjustment of the frame capture rate during periods of inactivity? 

A. There is an option in the Spike2 video application for the user to set a Slow Frame Rate that can 
be used during periods of inactivity to help minimise file size. Set up the slow and maximum frame 
rate to capture from the Set Slow Frame Rate option in the Settings menu of the Spike2 video 
window. You can then enable and disable the Slow Frame Rate mode using the associated 
toolbar button.  

   


'$ToneThreshold|Script to play a short tone in response to a threshold crossing on-line.

Var Chan%;			'Channel to monitor
VAr Data%;			'Handle of data file
Var Snd%:=0;		'Flag for sound on/off
Var Num%;			'HCursor no.
Var dir%;			'Monitor data above/below threshold
Var ok%;				'OK for dialog
var gFloat%[20];              'global for floating window states

HideAll();		'Hide windows and toolbars

Data%:= FileNew(0,1);	'Open new data file ready to sample
if data%< 0 then			'If unable to open file
Message("Unable to open a data file! 1401 not switched on?");	'Warn user
RestoreAll();		'Restore windows & toolbars
halt;endif;			'and quit the script

DlgCreate("Set channel");		
DlgChan(1,"Set channel to monitor",1);	'Set channel to monitor 
ok%:=DlgShow(chan%);

if ok%< 1 then
FileClose(-1,-1);		'If user presses Cancel
RestoreAll();			'Restore windows & toolbar
halt;endif;				'Quit the script

Hcursordelete(-1);	'Clear any HCursors
Num%:=HcursorNew(chan%);		'Set a new HCursor on selected channel
HCursorLabel(4,num%,"Threshold");	'Label it

Interact("Set HCursor as Threshold level",1023);	'User positions threshold
Chan%:=HcursorChan(Num%);		'Get current channel for threshold if changed

DoToolbar();   'Create script toolbar
Halt;

Func DoToolbar()    
ToolbarClear();     'Remove any old buttons
ToolbarSet(0, "", Idle%);   'Idle routine
ToolbarSet(1, "&Quit", Quit%);    'This quits the script
ToolbarSet(2, "Sound On", Sound%);   'Sound ON/OFF
ToolbarSet(4, "Start Sampling", start%);	'Start sampling
ToolbarSet(5, "Stop Sampling", stop%);		'Stop sampling
ToolbarEnable(5,0);
return Toolbar("Press Sound to BEEP on or off", 1023);
end;

Func Idle%()   'Idle routine runs when system has time
Var Value;
If Snd%=1 then		'If sound ON
	Value:=ChanValue(chan%,Maxtime());		'Get current value of data
	If Value>Hcursor(Num%) and dir%=0 then;		'If data above threshold
		Sound(127,0.05);		'Sound the tone
		dir%:=1;		'Set flag for 'data above threshold'
	Endif
	If Value<Hcursor(num%) then	'If data below threshold
		dir%:=0;		'Set flag for 'data below threshold'
	Endif;
Endif;
return 1;  'This leaves toolbar active
end;

Func Quit%()
RestoreAll();	'Restore hidden toolbars and windows before quitting the script
return 0;
end;

Func Sound%()   'Sound ON/OFF
If Snd%=0 then 
	Snd%:=1;		'Set sound flag to ON	
	Toolbarset(2,"Sound Off", Sound%);		'Change button label
	else
	Snd%:=0;		'Set sound flag to OFF
	Toolbarset(2,"Sound On", Sound%);		'Change button label
Endif;
return 1;  'This leaves toolbar active
end;

func Start%()							'If "Start" is pressed

SampleStart();						'Start sampling
ToolbarEnable(4,0);				'Disable Sample start button
ToolbarEnable(5,1);				'Enable Sample stop button	
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()						'If "Stop" is pressed
SampleStop();						'Stop sampling
if (data%>0) and              'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(5,0);				'Disable Sample stop button
ToolbarText("Sampling stopped");
return 1;								'Stay in toolbar
end;


proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end



Threshold Beep.s2s


'SampleBeep.sgs|Signal script to play a beep at the start of each frame

var data%;								'Data file being sampled
var sFlag%;								'Flag to say if sound made
var mTime;								'Time at end of frame
var soundFreq%;						'MIDI value of beep frequency


'Set up toolbar
ToolbarSet(1, "Quit");				
ToolbarSet(3, "New file", NewFile%);
ToolbarSet(4, "Start Sampling", Start%);
ToolbarEnable(4,0);
Toolbar("Beep on start of frame",1023);


'Open new file for sampling
Func NewFile%()
data% := FileNew(0,2);		'Open new file based on current sampling configuration
if data% > 0 then			'If file opened successfully
	Window(0,0,100,100);	'Maximise file
	XRange(MinTime(),MaxTime());	'Scale X-axis
	mTime := XHigh();		
	FrontView(data%);		'Make sure data is current view
	ToolbarEnable(4,1);		'Enable Start button
	ToolbarEnable(3,0);		'Disable New button
	SamplePause(1);							'Set pause at end of sweep
else
	Message("Failed to create new data file\nCheck 1401 is connected and retry");
endif;
return 1;
end;


'Start sampling
Func Start%()
View(data%);
SampleStart();		'Start sampling
ToolbarSet(0,"Idle",Idle%);		'Sey Idle routine to monitor data
ToolbarClear(4);				
ToolbarSet(4, "Stop", Stop%);	'Set Stop button
return 1;
end;


'Stop sampling
Func Stop%()
View(data%);
SampleStop();		'Stop sampling
ToolbarClear(4);	
ToolbarSet(4, "Start", Start%);		'Set Start button
ToolbarEnable(4,0);		'Disable Start button	
ToolbarEnable(3,1);		'Enable New file button
ToolbarClear(0);
return 1;			'Clear Idle routine
end;


'Idle routine to monitor when sweep is taking place and "beep"
Func Idle%()									
View(data%);
if sFlag% = 0 then							'If flag not set
	if SampleStatus() = 2 then				'If sweep in progress
		if MaxTime(1) > MinTime() and MaxTime(1) < (mTime - BinSize(1)) then			'Check to see that some data has been sampled
			soundFreq% := (FrameState(0)*6) + 49;				'Set beep frequency based on frame state
			Sound(soundFreq%,0.05,1);		'Make sound				
			sFlag% := 1;						'Set flag to say sound made
		endif;
	endif;
else
	if SampleStatus() = 3 then 			'If paused at end of sweep
		sFlag% := 0;							'Clear flag
		SampleSweep();							'Release pause to prepare for next sweep
	endif;
endif;
return 1;
end;


SampleBeep.sgs

If you have any comments about the newsletter format and content, or wish to un-subscribe from the 
mailing list for this newsletter, please notify sales@ced.co.uk. 

